
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of Rel-13 eMTC test case 7.1.4a.1 which is part of the LTE-A PRO test suite in the ‘iwd-TTCN3-B2017-06_D17wk51’ ATS delivery.

The test case can be demonstrated to run with two different Rel-13 LTE-A PRO UEs (see section 5). Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
7.1.4a.1
Test Group:
LTE_A_PRO\7_1\MAC_eMTC.ttcn
ATS Version:
iwd-TTCN3-B2017-06_D17wk51
System Simulator used:
R&S CMW500




UE used:
Qualcomm MDM9207, Spreadtrum RDA8910
Verification Status:
PASS


4. Corrections required for eMTC TC 7.1.4a.1
Change 1  – Correction to function ‘f_TC_7_1_4a_1_EUTRA’

	Function name
	f_TC_7_1_4a_1_EUTRA ()

	Reason for change
	1.   The start PRB is not properly configured (i.e. does not correspond to the system bandwidth), when assigning uplink grants to the UE.
2.   The System Information modification is not indictaed and the SI value tag is not incremented in System Information Block Type 1-BR. Moreover, the UE is not allowed enough time to read the modified System Information based on BCCH modification period.
3.   Inside function ‘fl_StepsforTP1_2’, the DL assignments are configured for repetitions and therefore have to be reset to default after the function call.

4.   For Test Loop #3, the uplink PUSCH repetitions are enabled and also the PUSCH DATA sent on UL is always with the same sequence number. Which means that the lower layers will perform duplicate discard for the repetitions. TTCN will therefore not receive the uplink data for the second iteration of the loop and the TC fails.
5.   For Test Loop #3, DCI for uplink grant assignment has to be scheduled on subframe 9.

6.   As specifc uplink grants are assigned in Test Loop #3, the default uplink grant assignments has to be stopped before calling ‘fl_StepsforTP3’.

	Summary of change
	1.   The start PRB is configured corresponding to the uplink bandwidth, when assigning uplink grants to the UE.
2.   The modified System Information is indicated by Direct Information indication and also the SI value tag is incremented in Sysem Information Block Type 1-BR. Moreover, a delay of 15s has been added at SS side, to allow the UE reading the modified System Information based on BCCH modification period.
3.   Added a function call to re-configure default DL assignments after function fl_StepsforTP1_2 was executed.
4.   For Test Loop# 3, TM mode is activated before function call ‘fl_StepsforTP3’ and deactivated before the postamble. 
5.   DCI for uplink grant is scheduled on subframe 9 for Test Loop #3.
6.   The default uplink grant assignments are stopped before calling function ‘fl_StepsforTP3’.

	TTCN module
	LTE_A_PRO\7_1\MAC_eMTC.ttcn

	MCC160 Comment
	


Before  change

...

function f_TC_7_1_4a_1_EUTRA() runs on EUTRA_PTC
  {
    /* L2/MAC
     * 7.1.4a.1 : Correct downlink reception and uplink transmission when specific valid 
    subframes is signaled for BL UE
     */
    var integer  v_RBStart :=0;
    var integer  v_UL_RBStart :=1;
    var integer  v_Imcs :=10;  //For CE mode A with Nprb =1, this results in 144 bits grant
    var integer v_Nprb := 1;
    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;
    var Dl_Bandwidth_Type v_Dl_Bandwidth;
    var Combination_Type  v_Combination;
    var  PdcchDciFormat_Type v_PdcchDciFormat := dci_6_1A;
    var  PdcchResourceAllocation_Type v_PdcchResourceAllocation := ra_2_Localised;
    var integer v_Subframe := 0; //suitable for FDD/tdd
    var template (value) Fdd_DownlinkOrTddSubframeBitmapBR_Type 
    v_Fdd_DownlinkOrTddSubframeBitmapBR:= cs_Fdd_DownlinkOrTddSubframeBitmapBR_10FDDTDD;
    ///////////////////////////////////////////////////////////////////////////////////
    //      PREAMBLE
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_Init (c1);
    v_Dl_Bandwidth := f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1);
    v_Combination  := f_EUTRA_CellInfo_GetCombination(eutra_Cell1);
    v_EUTRA_FDD_TDD_Mode :=f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);
    //@sic R5-172502 sic@
    if (pc_FDD_TypeB_HalfDuplex) {
      v_Subframe := 8;
      v_Fdd_DownlinkOrTddSubframeBitmapBR:= cs_Fdd_DownlinkOrTddSubframeBitmapBR_10HalfDupFDD;
    }
    if (f_GetTestcaseAttrib_CEModeB(testcasename())) {
      v_PdcchDciFormat := dci_6_1B;
      v_PdcchResourceAllocation := ra_Unspecified;
      v_Nprb := 4;
      v_Imcs := 1;
    }
    select (v_Dl_Bandwidth) {
      case (n25) {   // 5 Mhz
        v_RBStart:= 19;
      }
      case (n50) {  //10 Mhz
        v_RBStart := 37; 
      }
      case (n100) { //20 MhZ
        v_RBStart := 86;
      }
    }
    // creates cell and performs registration and moves to State 2A-CE
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);
    // Disable AS ciphering on UE side.
    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm
    ///////////////////////////////////////////////////////////////////////////////////
    //      Test Body
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_TestBody_Set(true, layer2);
    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now, 
     cs_DciInfo_CcchDcchDtchUL_Def);
    //@siclog "Step 1-3" siclog@
    f_EUTRA_CellInfo_SetSysInfo_Fdd_DownlinkOrTddSubframeBitmapBR(eutra_Cell1,
    v_Fdd_DownlinkOrTddSubframeBitmapBR);
    f_EUTRA_ModifySysinfo(eutra_Cell1, false);
    //@siclog "Step 3" siclog@
    //Wait for 15s for the UE to receive system information.
    f_Delay(15.0);
    //@siclog "Step 4" siclog@
    f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling();
    //@siclog "Steps 5-20 " siclog@
    fl_StepsforTP1_2(v_PdcchDciFormat,v_PdcchResourceAllocation,v_Imcs,v_Nprb,v_RBStart,v_Subframe);
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1,
    cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1, cs_TimingInfo_Now,                                                       
    cs_DciDLInfo_Auto(cs_DciInfo_CcchDcchDtchDL(v_Dl_Bandwidth))));
    //@siclog "Step 21" siclog@
    f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);
    //@siclog "Step 22" siclog@
    f_Delay (5.0);
    //@siclog "Step 23-25" siclog@
    //@sic R5-172502 sic@
    if (pc_FDD_TypeB_HalfDuplex) {
      v_Fdd_DownlinkOrTddSubframeBitmapBR:= cs_Fdd_DownlinkOrTddSubframeBitmapBR_40HalfDupFDD;
    }
    else {
      v_Fdd_DownlinkOrTddSubframeBitmapBR:= cs_Fdd_DownlinkOrTddSubframeBitmapBR_40FDDTDD;
    }
   f_EUTRA_CellInfo_SetSysInfo_Fdd_DownlinkOrTddSubframeBitmapBR(eutra_Cell1,
   v_Fdd_DownlinkOrTddSubframeBitmapBR);
    f_EUTRA_ModifySysinfo(eutra_Cell1, false);  
    //Wait for 15s for the UE to receive system information.
    f_Delay(15.0);
    //@siclog "Step 26" siclog@
    f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling();
    //@siclog "Steps 27-42 " siclog@
    fl_StepsforTP1_2(v_PdcchDciFormat,v_PdcchResourceAllocation,v_Imcs,v_Nprb,v_RBStart,v_Subframe);
    // configure SS in normal grant transmission
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1,
    cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1,cs_TimingInfo_Now, 
    cs_DciDLInfo_Auto(cs_DciInfo_CcchDcchDtchDL(v_Dl_Bandwidth))));
    if (v_EUTRA_FDD_TDD_Mode == FDD) {
      //@siclog "Step 43" siclog@
      f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);
      //@siclog "Step 44" siclog@
      f_Delay (5.0);
      //@siclog "Step 45-47" siclog@
      f_EUTRA_CellInfo_SetSysInfo_Fdd_DownlinkOrTddSubframeBitmapBR(eutra_Cell1,omit); 
      f_EUTRA_CellInfo_SetSysInfo_FDD_UplinkSubframeBitmapBR(eutra_Cell1,
      cs_FDD_UplinkSubframeBitmapBR);
         f_EUTRA_ModifySysinfo(eutra_Cell1, false);         
        //Wait for 15s for the UE to receive system information.
         f_Delay(15.0);
         //@siclog "Step 48" siclog@
         f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling();
         if (f_GetTestcaseAttrib_CEModeB(testcasename()))
         {
            v_Imcs := 3;
         }
         //@siclog "Steps 49-60 " siclog@
         fl_StepsforTP3(v_Imcs,1,v_UL_RBStart,6); // Results in PUSCH in subframe 0
     }
    f_EUTRA_TestBody_Set(false);
    ///////////////////////////////////////////////////////////////////////////////////
    //      Postamble
    //////////////////////////////////////////////////////////////////////////////////
    f_DelayForDLPDU_Tx() ; //Delay of 100 ms for RLC status PDU transmission
    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble(eutra_Cell1, E3_TESTMODE);
  }
...

After  change

...

function f_TC_7_1_4a_1_EUTRA() runs on EUTRA_PTC
  {
    /* L2/MAC
     * 7.1.4a.1 : Correct downlink reception and uplink transmission when specific valid 
    subframes is signaled for BL UE
     */
    var integer  v_RBStart :=0;
    var integer  v_UL_RBStart :=1;
    var integer  v_Imcs :=10;  //For CE mode A with Nprb =1, this results in 144 bits grant
    var integer v_Nprb := 1;
    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;
    var Dl_Bandwidth_Type v_Dl_Bandwidth;
    var Combination_Type  v_Combination;
    var  PdcchDciFormat_Type v_PdcchDciFormat := dci_6_1A;
    var  PdcchResourceAllocation_Type v_PdcchResourceAllocation := ra_2_Localised;
    var integer v_Subframe := 0; //suitable for FDD/tdd
    var template (value) Fdd_DownlinkOrTddSubframeBitmapBR_Type 
    v_Fdd_DownlinkOrTddSubframeBitmapBR:= cs_Fdd_DownlinkOrTddSubframeBitmapBR_10FDDTDD;
    ///////////////////////////////////////////////////////////////////////////////////
    //      PREAMBLE
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_Init (c1);
    v_Dl_Bandwidth := f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1);
    v_Combination  := f_EUTRA_CellInfo_GetCombination(eutra_Cell1);
    v_EUTRA_FDD_TDD_Mode :=f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);
    //@sic R5-172502 sic@
    if (pc_FDD_TypeB_HalfDuplex) {
      v_Subframe := 8;
      v_Fdd_DownlinkOrTddSubframeBitmapBR:= cs_Fdd_DownlinkOrTddSubframeBitmapBR_10HalfDupFDD;
    }
    if (f_GetTestcaseAttrib_CEModeB(testcasename())) {
      v_PdcchDciFormat := dci_6_1B;
      v_PdcchResourceAllocation := ra_Unspecified;
      v_Nprb := 4;
      v_Imcs := 1;
    }
    select (v_Dl_Bandwidth) {
      case (n25) {   // 5 Mhz
        v_RBStart:= 19;
        v_UL_RBStart := 0;
      }
      case (n50) {  //10 Mhz
        v_RBStart := 37; 
        v_UL_RBStart := 1;
      }
      case (n100) { //20 MhZ
        v_RBStart := 86;
        v_UL_RBStart := 2;
      }
    }
    // creates cell and performs registration and moves to State 2A-CE
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);
    // Disable AS ciphering on UE side.
    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm
    ///////////////////////////////////////////////////////////////////////////////////
    //      Test Body
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_TestBody_Set(true, layer2);
    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now, 
     cs_DciInfo_CcchDcchDtchUL_Def);
    //@siclog "Step 1-3" siclog@
    f_EUTRA_CellInfo_SetSysInfo_Fdd_DownlinkOrTddSubframeBitmapBR(eutra_Cell1,
    v_Fdd_DownlinkOrTddSubframeBitmapBR);
    f_EUTRA_SetSB1_ValueTag(eutra_Cell1, true);
    f_EUTRA_ModifySysinfo_ValueTagBR(eutra_Cell1, true, -, true); 
    //@siclog "Step 3" siclog@
    //Wait for 15s for the UE to receive system information.
    f_Delay(15.0);
    //@siclog "Step 4" siclog@
    f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling();
    //@siclog "Steps 5-20 " siclog@
    fl_StepsforTP1_2(v_PdcchDciFormat,v_PdcchResourceAllocation,v_Imcs,v_Nprb,v_RBStart,v_Subframe);
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1,
    cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1, cs_TimingInfo_Now,                                                       
    cs_DciDLInfo_Auto(cs_DciInfo_CcchDcchDtchDL(v_Dl_Bandwidth))));
    //@siclog "Step 21" siclog@
    f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);
    //@siclog "Step 22" siclog@
    f_Delay (5.0);
    //@siclog "Step 23-25" siclog@
    //@sic R5-172502 sic@
    if (pc_FDD_TypeB_HalfDuplex) {
      v_Fdd_DownlinkOrTddSubframeBitmapBR:= cs_Fdd_DownlinkOrTddSubframeBitmapBR_40HalfDupFDD;
    }
    else {
      v_Fdd_DownlinkOrTddSubframeBitmapBR:= cs_Fdd_DownlinkOrTddSubframeBitmapBR_40FDDTDD;
    }
   f_EUTRA_CellInfo_SetSysInfo_Fdd_DownlinkOrTddSubframeBitmapBR(eutra_Cell1,
   v_Fdd_DownlinkOrTddSubframeBitmapBR);
    f_EUTRA_SetSB1_ValueTag(eutra_Cell1, true);
    f_EUTRA_ModifySysinfo_ValueTagBR(eutra_Cell1, true, -, true); 
    //Wait for 15s for the UE to receive system information.
    f_Delay(15.0);
    //@siclog "Step 26" siclog@
    f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling();
    //@siclog "Steps 27-42 " siclog@
    fl_StepsforTP1_2(v_PdcchDciFormat,v_PdcchResourceAllocation,v_Imcs,v_Nprb,v_RBStart,v_Subframe);
    // configure SS in normal grant transmission
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1,
    cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1,cs_TimingInfo_Now, 
    cs_DciDLInfo_Auto(cs_DciInfo_CcchDcchDtchDL(v_Dl_Bandwidth))));
    if (v_EUTRA_FDD_TDD_Mode == FDD) {
      //@siclog "Step 43" siclog@
      f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);
      //@siclog "Step 44" siclog@
      f_Delay (5.0);
      //@siclog "Step 45-47" siclog@
      f_EUTRA_CellInfo_SetSysInfo_Fdd_DownlinkOrTddSubframeBitmapBR(eutra_Cell1,omit); 
      f_EUTRA_CellInfo_SetSysInfo_FDD_UplinkSubframeBitmapBR(eutra_Cell1,
      cs_FDD_UplinkSubframeBitmapBR);
         //WI=339028
         f_EUTRA_SetSB1_ValueTag(eutra_Cell1, true);
         f_EUTRA_ModifySysinfo_ValueTagBR(eutra_Cell1, true, -, true);
         //Wait for 15s for the UE to receive system information.
         f_Delay(15.0);
         //@siclog "Step 48" siclog@
         f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling();
         f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1,  

         cs_MAC_TestMode_NoHeaderManipulationDL_UL);
         f_EUTRA_StopULGrantTransmission_No_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);
         if (f_GetTestcaseAttrib_CEModeB(testcasename()))
         {
            v_Imcs := 3;
         }
         //@siclog "Steps 49-60 " siclog@
         fl_StepsforTP3(v_Imcs,1,v_UL_RBStart,9); // Results in PUSCH in subframe 0
     }
    f_EUTRA_TestBody_Set(false);
    ///////////////////////////////////////////////////////////////////////////////////
    //      Postamble
    //////////////////////////////////////////////////////////////////////////////////
    f_DelayForDLPDU_Tx() ; //Delay of 100 ms for RLC status PDU transmission
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NormalMode);
    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble(eutra_Cell1, E3_TESTMODE);
  }
...

Change 2  – Correction to function ‘fl_stepsforTP3’

	Function name
	fl_StepsforTP3 ()

	Reason for change
	1.   There are no predefind variables to handle PUSCH repetitions.
2.   For Test Loop #3, the downlink PDU has to be generated with a different sequence number for sending the PDU in TM mode.
3.   For Test Loop #3, PUSCH repetitions have to be handled explicitly.
4.   Transmission of RLC status PDU has to be explicitly handled.

	Summary of change
	1.   Defind a variable “v_Num_UL_Repetition” and assigned the value based on variable “i”. Further used this variable “v_Num_UL_Repetition” in the loop for DRB receive indication. 
2.   For Test Loop #3, variables “v_EncodedPDCPPDU” and “v_EncodedRLCPDU” are defined and assigned with values for the encoded data. Further these variable “v_EncodedRLCPDU” are used in  “DRB_COMMON_REQ”.
3.   For Test Loop #3, variable “j” is defined with a value that corresponds to the expected PUSCH uplink data repetitions for each loop. Further used this variable in the loop for handling the repetitions. Defined variable “DRB_COMMON_IND” and used it in DRB receive indication.

4.   Added a function for transmission of RLC ststaus PDU. Moreover, the step numbers in verdict assignments have been corrected.


	TTCN module
	LTE_A_PRO\7_1\MAC_eMTC.ttcn

	MCC160 Comment
	


Before  change

...

function fl_StepsforTP3(integer             p_Imcs,
                          integer             p_Nprb,
                          integer             p_RBStart,
                          integer             p_SubFrame) runs on EUTRA_PTC
  {
    var SubFrameTiming_Type v_Timing;
    var  RepetitionNumber_Type v_RepetitionNumber := 1;
    var integer i;
    timer t_Watchdog := 5.0;
    for (i :=0; i < 4; i:= i+1) {
      v_RepetitionNumber := i;
      //@siclog "Step 49-52-55-58" siclog@
      v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1, 200);
      DRB.send (cas_DRB_COMMON_REQ_PDCP_SDU (eutra_Cell1,tsc_RbId_DRB1,
      cs_TimingInfo(v_Timing), cs_PDCP_SDUList_1 (crs_PDCP_SDU_10B)));
      // one time UL grant
      v_Timing:=f_EUTRA_NB_TimingInfoAdd(v_Timing,60);
      v_Timing.Subframe.Number := p_SubFrame;
      //@siclog "Step 50-53-56-59" siclog@
      //Send UL Grant
      f_EUTRA_OneULGrantTransmission(eutra_Cell1,
                                     cs_TimingInfo(v_Timing),
                                     p_Imcs,
                                     p_Nprb,
                                     -,
                                     -,
                                     -,
                                     p_RBStart,
                                     v_RepetitionNumber);
      //@siclog "Step 51-54-57-60" siclog@
      t_Watchdog.start;
      DRB.receive (car_DRB_COMMON_IND_PDCP_SDUList (eutra_Cell1, tsc_RbId_DRB1, ?,     

      cr_PDCP_SDUList_1(crs_PDCP_SDU_10B)));
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8-12-16-20");
      t_Watchdog.stop;
    }
  }
...

After  change

...

function fl_StepsforTP3(integer             p_Imcs,
                          integer             p_Nprb,
                          integer             p_RBStart,
                          integer             p_SubFrame) runs on EUTRA_PTC
  {
    var SubFrameTiming_Type v_Timing;
    var  RepetitionNumber_Type v_RepetitionNumber := 1;
    var  integer  v_Num_UL_Repetition := 0;
    var  integer  j := 0;
    var DRB_COMMON_IND v_DRB_COMMON_IND;
    var octetstring v_EncodedPDCPPDU;
    var octetstring v_EncodedRLCPDU;
    var integer i;
    timer t_Watchdog := 5.0;
    for (i :=0; i < 4; i:= i+1) {
      v_RepetitionNumber := i;
      if (i==0){
      

v_Num_UL_Repetition := 1;
      }else if (i==1){
      

v_Num_UL_Repetition := 2;
      }else if (i==2){
      

v_Num_UL_Repetition := 4;
      }else if (i==3){
      

v_Num_UL_Repetition := 8;
      }
      //@siclog "Step 49-52-55-58" siclog@
      v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1, 200);
      v_EncodedPDCPPDU := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN
      (cs_PDCP_DataPdu_LongSN(int2bit(i, 12), crs_PDCP_SDU_13B)));
      v_EncodedRLCPDU  := f_RLC_AMD_1PDU_Encvalue(i, tsc_NoPoll, v_EncodedPDCPPDU);

      DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(eutra_Cell1,tsc_RbId_DRB1,
      cs_TimingInfo(v_Timing), cs_MAC_PDUList_1 
      (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRLCPDU)))); 
      // one time UL grant
      v_Timing:=f_EUTRA_NB_TimingInfoAdd(v_Timing,60);
      v_Timing.Subframe.Number := p_SubFrame;
      //@siclog "Step 50-53-56-59" siclog@
      //Send UL Grant
      f_EUTRA_OneULGrantTransmission(eutra_Cell1,
                                     cs_TimingInfo(v_Timing),
                                     p_Imcs,
                                     p_Nprb,
                                     -,
                                     -,
                                     -,
                                     p_RBStart,
                                     v_RepetitionNumber);
      //@siclog "Step 51-54-57-60" siclog@
      t_Watchdog.start;
      for (j:=0; j < v_Num_UL_Repetition; j:= j+1) {
        //@siclog "Steps 5 - 7 - 9" siclog@
        //@siclog "Step 14 void" siclog@
       alt {
             []DRB.receive (car_DRB_COMMON_IND_MAC            

             (eutra_Cell1,tsc_RbId_DRB1,cr_TimingInfo ( ?, (2,3)), 
             {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet18_Any)})) ->
             value v_DRB_COMMON_IND
            {
              f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 51-54-57-60");
            }
        }
      }
      f_RLC_Status_PDU_Tx (tsc_LchId_DRB1, i+1);
      t_Watchdog.stop;
    }
  }

}
...

Change 3 – Correction to various templates
	Template name
	cs_Fdd_DownlinkOrTddSubframeBitmapBR_10HalfDupFDD,
cs_Fdd_DownlinkOrTddSubframeBitmapBR_40HalfDupFDD,

cs_FDD_UplinkSubframeBitmapBR

	Reason for change
	1. The values assigned to the templates cs_Fdd_DownlinkOrTddSubframeBitmapBR_10HalfDupFDD and cs_Fdd_DownlinkOrTddSubframeBitmapBR_40HalfDupFDD are not according to the values specified in prose (TS 36.523-1).
2. As per TS 36.523-3 clause 7.26, UL assignments are sent on subframe 8, so PUSCH is expected on subframe 2. Therefore, when signalling SystemInformationBlockType1-BR-r13 in step 45 (= 3rd Test Loop), subframe 2 should not be restricted / disabled for UL transmission in the respective ‘FDD-UplinkSubframeBitmapBR-r13’ IE. (( see Prose CR) 

	Summary of change
	1. The values assigned to the templates cs_Fdd_DownlinkOrTddSubframeBitmapBR_10HalfDupFDD and cs_Fdd_DownlinkOrTddSubframeBitmapBR_40HalfDupFDD are corrected according to the values specified in prose (TS 36.523-1).
2. The values assigned to the template cs_FDD_UplinkSubframeBitmapBR  are corrected in order to allow UL transmission in subframe 2. (( see Prose CR)

	TTCN module
	LTE_A_PRO\7_1\MAC_eMTC.ttcn

	MCC160 Comment
	


Before  change

...

template (value) Fdd_DownlinkOrTddSubframeBitmapBR_Type cs_Fdd_DownlinkOrTddSubframeBitmapBR_10FDDTDD :=
  {
    subframePattern10_r13 := '1000000001'B //@sic R5s172502 R5-176809 sic@
  };
  template (value) Fdd_DownlinkOrTddSubframeBitmapBR_Type cs_Fdd_DownlinkOrTddSubframeBitmapBR_10HalfDupFDD :=
  {
    subframePattern10_r13 := '1000000011'B  //@sic R5-176809 sic@
  };
  template (value) Fdd_DownlinkOrTddSubframeBitmapBR_Type cs_Fdd_DownlinkOrTddSubframeBitmapBR_40FDDTDD :=
  {
    subframePattern40_r13 := '1000000000100000000010000000001000000000'B
  };
  template (value) Fdd_DownlinkOrTddSubframeBitmapBR_Type cs_Fdd_DownlinkOrTddSubframeBitmapBR_40HalfDupFDD :=
  {
    subframePattern40_r13 := '1000000010100000001010000000101000000010'B
  };
  template (value) FDD_UplinkSubframeBitmapBR_Type  cs_FDD_UplinkSubframeBitmapBR := '1000000000'B
...

After  change

...

template (value) Fdd_DownlinkOrTddSubframeBitmapBR_Type cs_Fdd_DownlinkOrTddSubframeBitmapBR_10FDDTDD :=
  {
    subframePattern10_r13 := '1000000001'B //@sic R5s172502 R5-176809 sic@
  };
  template (value) Fdd_DownlinkOrTddSubframeBitmapBR_Type cs_Fdd_DownlinkOrTddSubframeBitmapBR_10HalfDupFDD :=
  {
    subframePattern10_r13 := '1100000011'B  //@sic R5-176809 sic@
  };
  template (value) Fdd_DownlinkOrTddSubframeBitmapBR_Type cs_Fdd_DownlinkOrTddSubframeBitmapBR_40FDDTDD :=
  {
    subframePattern40_r13 := '1000000000100000000010000000001000000000'B  };
  template (value) Fdd_DownlinkOrTddSubframeBitmapBR_Type cs_Fdd_DownlinkOrTddSubframeBitmapBR_40HalfDupFDD :=
  {
    subframePattern40_r13 := '1100000011110000001111000000111100000011'B
  };
  template (value) FDD_UplinkSubframeBitmapBR_Type  cs_FDD_UplinkSubframeBitmapBR := '0011000000'B
...

5. Execution Log Files

5.1 Qualcomm MDM9207
The Qualcomm MDM9207 UE passed this LTE-A-PRO (Rel-13) eMTC test case on R&S CMW500 Multi-RAT Protocol Tester in LTE FDD band 4.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
Rohde-Schwarz\tc_7_1_4a_1_LTE_FDD_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

5.1 Spreadtrum RDA8910
The Spreadtrum RDA8910 UE passed this LTE-A-PRO (Rel-13) eMTC test case on R&S CMW500 Multi-RAT Protocol Tester in LTE FDD band 1. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
Rohde-Schwarz\tc_7_1_4a_1_LTE_FDD_Spreadtrum.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

